

UYGULAMA KODLARI

```
#include <SPI.h>
// SPI haberleşmesi için gerekli kütüphane
#include <MFRC522.h>
// RFID RC522 modülü için kütüphane
#include <Servo.h>
// Servo motor kontrolü için kütüphane

#define SDA_PIN 10
// RFID modülünün SDA pini
#define RST_PIN 9 // RFID reset pini
#define SERVO_PIN 3 // Servo motor pini

MFRC522 rfid(SDA_PIN, RST_PIN);
// RFID nesnesi oluşturulur
Servo kapi; // Servo motor nesnesi

byte izinliKart[4] = {0x84, 0x70, 0xF8, 0x05};
// Yetkili kart UID bilgisi

unsigned long oncekiZaman = 0;
// Kapının açıldığı anın zamanı
const unsigned long kapiAcikSuresi = 3000;
// Kapının açık kalma süresi (ms)

bool kapiAcik = false;
// Kapının mevcut durumunu tutar

void setup() {
    SPI.begin(); // SPI haberleşmesini başlatır
    rfid.PCD_Init(); // RFID modülünü başlatır
    kapi.attach(SERVO_PIN); // Servo motoru bağlar
    kapiKapat();
    // Başlangıçta kapıyı kapalı konuma getirir
}

void loop() {
    kartKontrol(); // RFID kart kontrol işlemleri
    kapiZamanKontrol();
    // Kapının açık kalma süresini millis() ile kontrol eder
}
```

```
// ----- FONKSİYONLAR -----

void kartKontrol() {
    // RFID kart okunup okunmadığını kontrol eder
    if (!rfid.PICC_IsNewCardPresent()) return;
    // Yeni kart yoksa çık
    if (!rfid.PICC_ReadCardSerial()) return;
    // Kart okunamazsa çık

    if (yetkiliMi()) { // Kart yetkiliyse
        kapiAc(); // Kapıyı aç
    }

    rfid.PICC_HaltA(); // Kart ile iletişimi sonlandır
}

bool yetkiliMi() {
    // Okutulan kartın yetkili olup olmadığını kontrol eder
    for (byte i = 0; i < 4; i++) {
        if (rfid.uid.uidByte[i] != izinliKart[i]) {
            return false; // UID uyuşmazsa yetkisiz
        }
    }
    return true; // Tüm UID eşleşirse yetkili
}

void kapiAc() { // Kapıyı açan fonksiyon
    if (!kapiAcik) { // Kapı zaten açık değilse
        kapi.write(90); // Servo motoru açma açısına getir
        kapiAcik = true; // Kapı açık durumuna geç
        oncekiZaman = millis(); // Açılma zamanını kaydet
    }
}

void kapiKapat() { // Kapıyı kapatan fonksiyon
    kapi.write(0); // Servo motoru kapalı konuma getir
    kapiAcik = false; // Kapı kapalı durumuna geç
}

void kapiZamanKontrol() {
    // Kapının açık kalma süresini kontrol eder
    if (kapiAcik && millis() - oncekiZaman >= kapiAcikSuresi) {
        kapiKapat(); // Süre dolunca kapıyı kapat
    }
}
```